
Eine Auswahlliste mit VBA 

Excelstammtisch vom 28.04.2020 

(zusammengefasst von René Martin) 

1. Versionsunabhängig programmieren 

Da dieses Tool von verschiedenen Kunden in verschiedenen Excelplattformen verwendet wird, 
versuche ich versionsunabhängig zu programmieren. 

Mir ist aufgefallen, dass VBA manchmal (!) bei niedrigeren Versionen Fehler produziert: 
Funktionen der Bibliothek VBA werden nicht erkannt. Deshalb habe ich mir angewöhnt statt 
Left, Right, Len, InStr, … zu schreiben: VBA.Left, VBA.Right, VBA.Len, VBA.InStr, … 

 

An einer Stelle wollte ich eine Liste sortieren: 

Ich habe auf Klasse ArrayList zugegriffen: 

Private Sub cmdSort_Click() 
    Dim i As Integer 
    Dim listMatrix As Object, varDaten As Variant 
     
    Set listMatrix = CreateObject("System.Collections.ArrayList") 
     
    For i = 0 To Me.lstAuswahl.ListCount - 1 
        listMatrix.Add Me.lstAuswahl.List(i) 
    Next i 
 
    listMatrix.Sort 
    Me.lstAuswahl.Clear ' -- Auswahlliste leeren 
     
    For Each varDaten In listMatrix 
        Me.lstAuswahl.AddItem varDaten 
    Next varDaten 

    If Me.lstAuswahl.ListCount > 0 Then 
        Me.lstAuswahl.ListIndex = 0 
    End If 
     
    Set listMatrix = Nothing 
End Sub 

Allerdings ist ArrayList ein Objekt von .NET-Framework 3.5. Ist dieses nicht auf einem Rechner 
installiert, funktionieren die Befehle der Klasse auch nicht. 

 

Also – raus damit – und die klassische Variante: 

Private Sub cmdSort_Click() 
    Dim intLast As Integer, intNext As Integer 
    Dim strTemp As String 
    With Me.lstAuswahl 
        For intLast = 0 To .ListCount - 1 
            For intNext = intLast + 1 To .ListCount - 1 
                If .List(intLast) > .List(intNext) Then 
                    strTemp = .List(intLast) 
                    .List(intLast) = .List(intNext) 
                    .List(intNext) = strTemp 
                End If 
            Next intNext 
        Next intLast 
    End With 
End Sub 

Überhaupt vermeide ich early binding. Ich vermeide überhaupt jede Art des Einbindens (auch 
kein late binding). Der Kundenwusch war ein Steuerelement einzubinden für eine bequeme 
Datumsauswahl, wie wir sie sie aus anderen Programmen kennen. Meine Antwort lautete: „nein!“ 
Ich habe so etwas nachgebaut (abgeschaut bei Outlook): 

2. Der Aufbau des Interviewfragebogens 

Auf dem Interviewfragebogen befinden sich verschiedene Eingabefelder: 

 Freie Eingabefelder (beispielsweise Name des Verantwortlichen oder Kursbeschreibung) 
 fest vorgegebene Datenüberprüfungen: 

 

 Dropdownlisten, die ihre Werte aus dem Tabellenblatt tbl_Basisdaten holen: 

Dialoge, über die Daten eingetragen werden können: 

 

Den Dialog mit seinen Varianten kann man in Abschnitt 1 Versionsunabhängig programmieren 
sehen. 

 Zum Teil liegen hinter den Eingabefeldern bedingte Formatierungen: 

 

 Die Formatierungen beziehen sich auf die eingetragenen Werte oder auf Werte anderer 
Zellen: 

Oder einfache Verweisformeln, die eingetragene oder ausgewählte Texte wiederholen: 

Einige einfache Formeln runden das Formular ab: 

 

An einigen Stellen wird ein „x“ eingetragen. Dieses soll als Vorgabe in allen rechts danaben 
stehenden Zellen wiederholt werden. Auch das kann mit einer einfachen Formel realisiert 
werden. Diese ist überschreibbar: 

=WENN(UND(ZÄHLENWENN($E96:I96;"x")>0;K$93<>"");"x";"") 

3. Liste anzeigen 

Bei sehr vielen Einträgen ist es mühsam durch die Dropdownliste zu navigieren. Man kann weder 
suchen, noch den ersten Buchstaben eintragen, um zu dem entsprechenden Wert zu gelangen 
noch sortieren.  

 

Deshalb kamen wir auf die Idee einer Liste: 

Sie kann per Tastenkombination oder über das Symbol aufgerufen werden. 

4. Tastenkombination 

Ich persönlich gebe ungern Programme weiter, deren Elemente per Tastenkombination 
aufgerufen werden. Der Grund: man sieht den Shortcut nicht. Zwar kann ein Beschriftungstext 
darauf hinweisen; jedoch gibt es bequemere Möglichkeiten auf Funktionen zuzugreifen. 

Will man einen Shortcut in Excel programmieren, muss beim Start der Datei beim Eregnis 
Workbook_Open das Makro auf eine Tastenkombination gelegt werden: 

Private Sub Workbook_Open() 
    Application.OnKey "^q", "ListeAnzeigen"  ' Tastenkombination 
End Sub 

Beim Schließen sollte diese dann wieder geleert werden: 

Private Sub Workbook_BeforeClose(Cancel As Boolean) 
    Application.OnKey "^q", ""  ' Tastenkombination entfernen 
End Sub 

5. Datenüberprüfung prüfen 

Leider gibt es in Excel keine Möglichkeit per VBA festzustellen, ob hinter einer Zelle eine 
Datenüberprüfung liegt. Die einzige Möglichkeit: man muss so tun als liege eine 
Datenüberprüfung auf der Zelle und fragt nun nach dem Wert, also fragt die Art der 
Datenüberprüfung ab: 

    On Error Resume Next 
[…] 
        i = ActiveCell.Validation.Type 
        If Err.Number <> 0 Or i <> 3 Then 
            MsgBox "In der Zelle " & ActiveCell.Address & _ 
       " wurde keine Auswahlliste hinterlegt.", vbInformation, "BCM" 

Da auf dem Interviewblatt „normale“ Datenüberprüfungen vorhanden sein können (Nummer 3), 
muss dieser Wert abgefangen werden. Würde keine Datenüberprüfung auf der Zelle liegen, wäre 
ein Fehler die Folge. Dieser kann abgefangen werden. 

An einer anderen Stelle wurde dies erneut benötigt und deshalb als Funktion ausgelagert: 

Public Function GibtEsDatenÜberprüfung(Zelle As Range) As Boolean 
    Dim blnVorhanden As Boolean 
    Dim s As String 
     
    On Error Resume Next 
    Err.Clear 
    s = Zelle.Validation.Formula1 
    If Err.Number = 0 Then 
        blnVorhanden = True 
    Else 
        blnVorhanden = False 
    End If 
    Err.Clear 
    GibtEsDatenÜberprüfung = blnVorhanden 
End Function 

6. Datenüberprüfung zerlegen 

Da die Listen der Datenüberprüfung entweder „hart“ eingetragen wurden oder als Verweis auf 
einen Zellbereich, muss dies überprüft werden. 

Dann werden die Teile eingetragen: 

With ActiveCell.Validation 
    strElemente = .Formula1 
    If strElemente Like "*!*" Then 
        strBlatt = Split(strElemente, "!")(0) 
        strBlatt = VBA.Replace(strBlatt, "=", "") 
        strElemente = Split(strElemente, "!")(1) 
        Set xlBereich = _ 
            ThisWorkbook.Worksheets(strBlatt).Range(strElemente) 
        For i = 1 To xlBereich.Rows.Count 
            Me.lstAuswahl.AddItem xlBereich.Cells(i) 
        Next 
    Else ' -- bei "harten" Listen 
        For i = 0 To UBound(Split(strElemente, ";")) 
            Me.lstAuswahl.AddItem Split(strElemente, ";")(i) 
        Next i 

    End If 
End With 

7. Bequem aus der Liste auswählen 

Die Liste kann sortiert werden (siehe Abschnitt 1 Versionsunabhängig programmieren). Oder 
man kann in ein Textfeld einen oder mehrere Buchstaben eintragen, so dass der erste Eintrag 
ausgewählt wird: 

Private Sub txtSuchfeld_Change() 
    Dim i As Integer 
    If Me.txtSuchfeld.Value <> "" Then 
        For i = 0 To Me.lstAuswahl.ListCount - 1 
            If Me.lstAuswahl.List(i) Like _ 
               Me.txtSuchfeld.Value & "*" Then 
                Me.lstAuswahl.ListIndex = i  
' -- markiere den Eintrag, der mit dem eingetragenen Text beginnt 
                Exit Sub 
            End If 
        Next 
    End If 
End Sub 


